

Project: DRA FRONT END FILTER PROJECT

Title: Specification of TSQL

Ref: DS/FMU/FEF/021

Issue: Revision : 2.2

Date: 5 June 2016

Status: Approved

Type: Specification

Keywords:

Author:

<i>Name</i>	<i>Location</i>	<i>Signature</i>	<i>Date</i>
G. M. Prout	WIN01		

Authorisation for Issue:

<i>Name</i>	<i>Function</i>	<i>Signature</i>	<i>Date</i>
R.B. Jones	HAT Manager		

Abstract: The formal specifications of TSQL for the DRA front end filter project RSRE 1C/6130.

Distribution: HAT FEF File
Simon Wiseman

0 DOCUMENT CONTROL

0.1 Contents List

0	DOCUMENT CONTROL	2
0.1	Contents List	2
0.2	Document Cross References	2
0.3	Changes History	3
0.4	Changes Forecast	3
1	GENERAL	4
1.1	Scope	4
1.2	Introduction	4
2	PRELIMINARIES	4
2.1	Consistency Proof for <i>repState</i> and <i>absState</i>	4
2.2	Auxiliary Functions	5
3	TSQL STATE	6
4	FUNCTIONALITY OF THE TSQL QUERY PROCESSING	8
4.1	Representation and Abstraction Functions for <i>State_t</i>	8
4.2	The function <i>processQuery_t</i>	9
5	UPDATING A TSQL STATE	10
6	TSQL TRANSITION FUNCTION	10
6.1	Building a Transition Function	10
6.2	The TSQL Transition Function	11
7	INDEX	12

0.2 Document Cross References

- [1] DS/FMU/FEF/004. *Specification of SSQL Semantics I*. G.M. Prout, ICL Secure Systems, WIN01.
- [2] DS/FMU/FEF/005. *Specifications of hide and updateState*. G.M. Prout, ICL Secure Systems, WIN01.
- [3] DS/FMU/FEF/010. *Proof of Security (IIa)*. G.M. Prout, ICL Secure Systems, WIN01.
- [4] DS/FMU/FEF/014. *Specification of SSQL Semantics II*. G.M. Prout, ICL Secure Systems, WIN01.
- [5] DS/FMU/FEF/018. *Proposal for Phase 2*. G.M. Prout, ICL Secure Systems, WIN01.
- [6] DS/FMU/FEF/022. *SWORD Front End Architectural Model*. R.D. Arthan, ICL Secure Systems, WIN01.

0.3 Changes History

Issue 1.1 (5 February 1993) First draft.

Issue 1.4 (23 February 1993) New parent *fef021.lattice_bottom* moved to DS/FMU/FEF/003.

Issue Revision : 2.2 (5 June 2016) Final approved version.

Issue 2.2 Removed dependency on ICL logo font

Issue 2.3 Stopped it reusing variable names from another document which was causing some make targets to fail.

0.4 Changes Forecast

Changes may be necessary as a result of issues raised during Phase 2.

1 GENERAL

1.1 Scope

This document gives a formal specification of the TSQL semantics. It constitutes deliverable D9 of work package 3, as given in the Proposal for Phase 2, [5].

1.2 Introduction

We propose defining the semantics of TSQL as a subset of SSQL with all classes in the state, except those that are actually stored as data, set at the lowest possible classification, *lattice_bottom*, and all worths *sterling*. This means that the *hide* function of [2] will have no effect on a TSQL state. A TSQL transition function will be comprised of two components *processQuery_t* and *updateState_t*. We will define the function *processQuery_t*, which captures the functionality of the semantics of TSQL, to be the same as *processQuery*, specified in [4], except that it will be defined on TSQL states and will only return *Effects* that have all classifications at *lattice_bottom* and all worths *sterling*. The function *updateState_t* operates on the output from the function *processQuery_t* and essentially determines whether or not the query should succeed.

A formal specification of the SWORD front end architectural model is given in [6]. The TSQL transition function specified here is described in [6] as the TSQL query processor of the conventional DBMS.

2 PRELIMINARIES

The following ProofPower instructions set up the new theory *fef021* and set the context for the proof tools.

```
SML
|open_theory "fef024";
|(force_delete_theory "fef021" handle _ => ());
|new_theory "fef021";
|push_pc "hol";
```

2.1 Consistency Proof for *repState* and *absState*

We have included here the consistency proofs for *repState* and *absState* which were defined in Phase 1 in [3].

```
SML
|push_consistency_goalΓ repState¬;
|a(strip_asm_tac (simple⇒_match_mp_rule type_lemmas_thm state_type_def));
|a(∃_tacΓ(rep,abs)¬ THEN asm_rewrite_tac[]);
|a ∧_tac;
```

SML

```

| (* *** Goal "1" *** *)
| a(REPEAT strip_tac THEN_TRY asm_rewrite_tac[]);
| a(POP_ASM_T (ante_tac o app_fun_ruleΓ abs∇) THEN asm_rewrite_tac[]);
| (* *** Goal "2" *** *)
| a(REPEAT ∇_tac THEN ⇒_tac);
| a(REPEAT strip_tac THEN_TRY asm_rewrite_tac[]);
| a(POP_ASM_T (ante_tac o app_fun_ruleΓ rep∇) THEN asm_rewrite_tac[]);
| save_consistency_thmΓ repState∇(pop_thm());
| val fef021_repState_absState_def = get_specΓ repState∇;

```

We simplify one of the properties of the representation and abstraction functions.

SML

```

| val fef021_isState_lemma1 = all_∇_intro(nth 1 (strip_∧_rule(all_∇_elim fef021_repState_absState_def)));

```

HOL output

```

| fef021_isState_lemma1 = ⊢ isState r ⇒ repState (absState r) = r

```

2.2 Auxiliary Functions

A function that takes an *Item* and sets its worth to *sterling*.

HOL Constant

```

| item_sterling : Item → Item
|-----
|
|   ∇ i • item_sterling i = if isNullItem i then i
|                       else
|                       let v = destValuedItem i
|                       in ValuedItemItem(MkValuedItem sterling (VI_val v))

```

A function that takes a piece of data and sets its class to bottom and its worth to sterling.

HOL Constant

```

| class_bottom : Data → Data
|-----
|
|   ∇ d • class_bottom d = let i = Dat_item d
|                       in MkData lattice_bottom (item_sterling i)

```

Now a function that takes a relation between column number and data and sets all classifications to *lattice_bottom* and worths to *sterling*.

HOL Constant

```

| set_bottomd : (Num ↔ Data) → (Num ↔ Data)
|-----
|
|   ∇ nd • set_bottomd nd = {(n,d)|n ∈ Dom nd ∧ d = class_bottom (nd @ n)}

```

A similar function that takes a relation between column number and update and sets all classifications to *lattice_bottom* and worths to *sterling*.

HOL Constant

$$\mathbf{set_bottom}_u : (Num \leftrightarrow Update) \rightarrow (Num \leftrightarrow Update)$$

$$\begin{aligned} \forall nu \bullet set_bottom_u \ nu = \\ \{(n,u) \mid n \in Dom \ nu \wedge u = \\ \quad let \ u' = nu \ @ \ n \\ \quad in \\ \quad if \ isItem \ u' \ then \ ItemUpdate \ (item_sterling \ (destItem \ u')) \\ \quad else \ if \ isClass \ u' \ then \ ClassUpdate \ lattice_bottom \\ \quad else \ DataUpdate \ (class_bottom \ (destData \ u'))\} \end{aligned}$$

3 TSQL STATE

We define the TSQL state $State_t$ as a subset of the SSQL state, $State : Exp$ specified in [1], where all classifications in the state are set at *lattice_bottom* except for those classifications that are stored as data. All worths in the state are set at *sterling*.

The constant $State_{tS}$ is the set of everything of type $State : Exp$ where the required classifications are at *lattice_bottom* and worths at *sterling*.

HOL Constant

State_{tS} : State \mathbb{P}

$$\begin{aligned}
State_{tS} = \{st \mid & \forall dir \bullet \\
& dir \in Ran (repState\ st) \\
& \Rightarrow Dir_exist\ dir = lattice_bottom \\
& \wedge Dir_class\ dir = lattice_bottom \\
& \wedge \forall tab \bullet \\
& \quad tab \in Ran(Dir_tables\ dir) \\
& \quad \Rightarrow TS_class\ tab = lattice_bottom \\
& \quad \wedge TS_maxRow\ tab = lattice_bottom \\
& \quad \wedge \forall col \bullet \\
& \quad \quad col \in TS_colspecs\ tab \\
& \quad \quad \Rightarrow CS_min\ col = lattice_bottom \\
& \quad \quad \wedge CS_max\ col = lattice_bottom \\
& \quad \wedge \forall cc \bullet \\
& \quad \quad cc \in Ran(TS_cons\ tab) \\
& \quad \quad \Rightarrow CC_exist\ cc = lattice_bottom \\
& \quad \wedge \forall row \bullet \\
& \quad \quad row \in Elems\ (TS_rows\ tab) \\
& \quad \quad \Rightarrow R_exist\ row = lattice_bottom \\
& \quad \quad \wedge \forall data \bullet \\
& \quad \quad \quad data \in Ran(R_data\ row) \\
& \quad \quad \quad \Rightarrow Dat_class\ data = lattice_bottom \\
& \quad \quad \quad \wedge isValuedItem\ (Dat_item\ data) \\
& \quad \quad \quad \Rightarrow VI_worth(destValuedItem(Dat_item\ data)) \\
& \quad \quad \quad = sterling\}
\end{aligned}$$

We define the property required on the representation state.

HOL Constant

isState_t : State \rightarrow Bool

$$\forall s \bullet isState_t\ s \Leftrightarrow s \in State_{tS}$$

We demonstrate that the new type will be non-empty.

SML

```

| push_goal([],⌈∃ s : State • isStatet s⌋);
| a(rewrite_tac[get_spec⌈isStatet⌋]);
| a(∃_tac⌈absState{ }⌋);
| a(rewrite_tac[get_spec⌈StatetS⌋]);
| a(lemma_tac⌈isState{ }⌋);
| (* *** Goal "1" *** *)
| a(rewrite_tac[get_spec⌋isState⌋,get_spec⌋StateS⌋,⇒_def,functional_def,∩_def,↔_def]);
| (* *** Goal "2" *** *)
| a(asm_fc_tac[fe021_isState_lemma1]);
| a(asm_rewrite_tac[get_spec⌋Ran⌋]);

```

Now the new type $State_t$ is defined.

SML

```

| val state_t_type_def = new_type_defn(["state_t_type_def"],"State_t",[],pop_thm());

```

4 FUNCTIONALITY OF THE TSQL QUERY PROCESSING

The functionality of the semantics of TSQL will be captured by the function $processQuery_t$ which is similar to $processQuery$, specified in [4] except that it works on TSQL states, of type $State_t$, with user clearance at $lattice_bottom$. The effect of $processQuery_t$ is the same as that of $processQuery$ with all classifications returned (except for those that are actually stored as data) set at $lattice_bottom$.

4.1 Representation and Abstraction Functions for $State_t$

First the representation function, $repState_t$, and abstraction function, $absState_t$, for the type $State_t$.

HOL Constant

```

| repStatet : Statet → State;
| absStatet : State → Statet

```

```

| (∀ a • absStatet (repStatet a) = a)
| ∧ (∀ r • isStatet r ⇒ repStatet (absStatet r) = r)
| ∧ (∀ a1 a2 • repStatet a1 = repStatet a2 ⇔ a1 = a2)
| ∧ (∀ r1 r2 • (isStatet r1 ∧ isStatet r2) ⇒
|   (absStatet r1 = absStatet r2 ⇔ r1 = r2))
| ∧ (∀ s • isStatet (repStatet s))

```

We prove the consistency of $repState_t$ and $absState_t$ and retrieve their definitions with the consistency obligation satisfied.

SML

```

| push_consistency_goalΓ repStatet⊥;
| a(strip_asm_tac (simple⇒_match_mp_rule type_lemmas_thm statet-type-def));
| a(∃_tacΓ(rep,abs)⊥ THEN asm_rewrite_tac[]);
| a ∧_tac;

```

SML

```

| (* *** Goal "1" *** *)
| a(REPEAT strip_tac THEN_TRY asm_rewrite_tac[]);
| a(POP_ASM_T (ante_tac o app_fun_ruleΓ abs⊥) THEN asm_rewrite_tac[]);
| (* *** Goal "2" *** *)
| a(REPEAT ∀_tac THEN ⇒_tac);
| a(REPEAT strip_tac THEN_TRY asm_rewrite_tac[]);
| a(POP_ASM_T (ante_tac o app_fun_ruleΓ rep⊥) THEN asm_rewrite_tac[]);
| save_consistency_thmΓ repStatet⊥(pop_thm());
| val repStatet-absStatet-def = get_specΓ repStatet⊥;

```

4.2 The function $processQuery_t$

HOL Constant

processQuery_t : $Query \times State_t \rightarrow Effect \times Errors$

```

∀ q s • processQueryt (q, s) =
  let (ef, es) = processQuery(q, lattice_bottom, repStatet s)
  in
  let ef' =
    if isInsert ef
    then let (t, ndl) = destInsert ef
         in InsertEffect(t, Map set_bottomd ndl)
    else if isDelete ef then ef
    else if isUpdate ef
    then let (t, nnu) = destUpdate ef
         in UpdateEffect(t, {(n, nu) | n ∈ Dom nnu ∧ nu = set_bottomu(nnu @ n)})
    else let dll = destSelect ef
         in SelectEffect(Map (Map class_bottom) dll)
  in (ef', es)

```

5 UPDATING A TSQL STATE

HOL Constant

$$\mathbf{updateState}_t : (Effect \times Errors) \times State_t \rightarrow State_t \times (Data\ LIST\ LIST \times Errors)$$

$$\forall efes : Effect \times Errors; s : State_t$$

- $updateState_t(efes, s)$

=

$$let (s', (c, out)) = updateState(lattice_bottom, efes, repState_t\ s)$$

$$in (absState_t\ s', out)$$

6 TSQL TRANSITION FUNCTION

A TSQL transition function is to be built from the two components $updateState_t$ and $processQuery_t$. We first give abbreviation definitions for the types $Process_t$ of the $processQuery_t$ component and $Ustate_t$ of the $updateState_t$ component.

SML

$$| declare_type_abbrev("Process_t", [], \ulcorner : Query \times State_t \rightarrow Effect \times Errors \urcorner);$$

SML

$$| declare_type_abbrev("Ustate_t", [], \ulcorner : (Effect \times Errors) \times State_t \rightarrow State_t \times (Data\ LIST\ LIST \times Errors) \urcorner);$$

We also give an abbreviation definition for the type of TSQL state transition functions, tf_t .

SML

$$| declare_type_abbrev("tf_t", [], \ulcorner : Query \times State_t \rightarrow State_t \times (Data\ LIST\ LIST \times Errors) \urcorner);$$

6.1 Building a Transition Function

We define a function, $MkTf_t$, which builds a transition function from two components: a component of type $Process_t$ and a component of type $Ustate_t$. The resulting transition function updates the original state of the database by using the result of processing a query on the state of the database.

HOL Constant

$$\mathbf{MkTf}_t : Process_t \rightarrow Ustate_t \rightarrow tf_t$$

$$\forall p:Process_t; u:Ustate_t; q : Query; s : State_t$$

- $(MkTf_t\ p\ u)\ (q, s) = u(p(q, s), s)$

6.2 The TSQL Transition Function

A TSQL transition function is a transition function built from the two components *processQuery_t* and *updateState_t*.

HOL Constant

TSQLtf : tf_t

$TSQLtf = MkTf_t \text{ processQuery}_t \text{ updateState}_t$

7 INDEX

<i>absState_t</i>	8
<i>class_bottom</i>	5
<i>fef021_repState_absState_def</i>	5
<i>fef021</i>	4
<i>isState_t</i>	7
<i>item_sterling</i>	5
<i>MkTf_t</i>	10
<i>processQuery_t</i>	9
<i>Process_t</i>	10
<i>repState_t_absState_t_def</i>	9
<i>repState_t</i>	8
<i>set_bottom_d</i>	5
<i>set_bottom_u</i>	6
<i>State_{tS}</i>	7
<i>tf_t</i>	10
<i>TSQLtf</i>	11
<i>updateState_t</i>	10
<i>Ustate_t</i>	10